# Copyright (c) 2011 Nokia Corporation.

QT += core gui opengl

TARGET = flagmines
TEMPLATE = app
VERSION = 1.1.1


#INCLUDEPATH += ./src_qtopenglspritebatch
#INCLUDEPATH += src_general
#INCLUDEPATH += src_qtopenglspritebatchtest
INCLUDEPATH += pi_src
INCLUDEPATH += .

# Define lite - version to limit difficulty levels.
# DEFINES += LITE_VERSION

SOURCES += \
    SpriteBatch.cpp \
    QtOpenGLSpriteBatch.cpp \
    mygamewindow.cpp \
    main.cpp \
    pi_src/thegame.cpp \
    pi_src/thelevel.cpp \
    pi_src/themenu.cpp \
    pi_src/particlesystem.cpp


HEADERS += \
    SpriteBatch.h \
    QtOpenGLSpriteBatch.h \
    mygamewindow.h \
    pi_src/thegame.h \
    pi_src/thelevel.h \
    pi_src/themenu.h \
    pi_src/particlesystem.h


symbian: {
    # FOR ENABLING HARDWARE FLOATS
    MMP_RULES += "OPTION gcce -march=armv6"
    MMP_RULES += "OPTION gcce -mfpu=vfp"
    MMP_RULES += "OPTION gcce -mfloat-abi=softfp"
    MMP_RULES += "OPTION gcce -marm"

    DEFINES += Q_OS_SYMBIAN
    ICON = icon.svg
    # For orientation locking.
    LIBS += -lcone -leikcore -lavkon

    # Store settings.
    vendorinfo = \
         "%{\"Digia\"}" \
         ":\"Digia\""

    my_deployment.pkg_prerules = vendorinfo
    DEPLOYMENT += my_deployment

    TARGET.UID3 = 0x200345f3

    #TARGET.UID3 = 0xee69dfd2

    TARGET.EPOCSTACKSIZE = 0x14000
    TARGET.EPOCHEAPSIZE = 0x020000 0x800000
}





windows: {
    QT += multimedia

    message("The project contains desktop ogl")
    message($$INCLUDEPATH)
    message($$LIBS)

}


    # Harmattan specific
contains(MEEGO_EDITION,harmattan) {
    DATADIR = /usr/share
    DEFINES += Q_WS_MAEMO_6
    target.path = /opt/flagmines/bin
    desktopfile.files += qtc_packaging/debian_harmattan/$${TARGET}.desktop
    desktopfile.path = $$DATADIR/applications
    icon.files = flagmines.png
    icon.path = /usr/share/icons/hicolor/128x128/apps
    INSTALLS += target desktopfile icon

OTHER_FILES += \
    qtc_packaging/debian_harmattan/rules \
    qtc_packaging/debian_harmattan/README \
    qtc_packaging/debian_harmattan/copyright \
    qtc_packaging/debian_harmattan/control \
    qtc_packaging/debian_harmattan/compat \
    qtc_packaging/debian_harmattan/changelog

}




RESOURCES += \
    resources.qrc
